
Source Code
{;} DB2WP.WPM - converts dBase III format comma delimited ASCII

files to WordPerfect 5.0 merge format.~

{;} Copyright (c) 1988, Stuart Bloom~

{;} Permission is granted to distribute this macro and its
accompanying documentation on a FREE basis only.  No
fee may be charged for copying, shipping, or transmission
without the author's written consent.
Permission is specifically granted to distribute this
macro through on-line data base services.~

{;} The user assumes all risks associated with the use of this
macro and by using it agrees to this assumption.~

{;} Note: comments have been removed from the .WPM file to save
space.~

{LABEL}mainmenu~
{ASSIGN}4~Q~
{CHAR}4~{^]}Action: 1{^\} {^V}C{^Q}onvert dBase III file to WP Merge; 

{^]}2{^\} Change {^V}D{^Q}ate fields; {^]}0{^\} {^V}Q{^Q}uit:  ~
{CASE CALL}"{VAR 4}"~

"1"~convert~
"C"~convert~
"c"~convert~
"2"~dodates~
"D"~dodates~
"d"~dodates~
"0"~quit~
"Q"~quit~
"q"~quit~

~
{Home}{Home}{Up}
{DISPLAY ON}

{GO}mainmenu~

{LABEL}quit~
{QUIT}

{LABEL}convert~
{DISPLAY OFF}
{CALL}getfile~
{CALL}commas~
{Home}{Home}{Up}
{CALL}replace~
{Home}{Home}{Up}

{RETURN}

{LABEL}commas~ {;} replace commas in quotes with tabs~

DB2WP.WPM Source Code Listing -  - September 16, 1988



{;} (tabs act as placeholders)~
{ON NOT FOUND} {;} after the last quoted field~

{RETURN}~
{Search}"{Search} {;} find a quoted field~
{Block}" {;} block it~
{Replace}N {;} replace within the block~

,{Search} {;} if there are commas inside the quotes~
{Tab}{Search} {;} replace them with tabs~

{Goto}{Goto} {;} back to the beginning of the field~
{Search}"{Search} {;} move past the end of the field~

{GO}commas~ {;} loop~

{LABEL}replace~ {;} replace delimiter codes with Merge codes~
{Replace}N {;} delete extra spaces at end of lines~

{Enter}{Search}
{Enter}{Search}

{Home}{Home}{Up}
{Replace}N

{Enter}{Search} {;} replace [HRt]s~
{Merge R}{Enter} {;} with ^R[HRt]~

{Merge Codes}E {;} plus ^E~
{HPg}{Search} {;} plus [HPg]~

{Home}{Home}{Up}
{Replace}N

,{Search} {;} replace commas~
{Merge R}{Enter} {;} with ^R[HRt]~

{Search}
{Home}{Home}{Up}
{Replace}N {;} delete quotes~

"{Search}
{Search}

{Home}{Home}{Up}
{Replace}N {;} restore commas inside fields~

{Tab}{Search} {;} tab placeholders are replaced~
,{Search} {;} with commas~

{RETURN}

{LABEL}dodates~
{DISPLAY OFF}
{CALL}getfile~ {;} user verifies right file is diaplayed~
{CALL}delimiter~ {;} user chooses delimiter character~
{CALL}year~ {;} user chooses year format~
{Home}{Home}{Down} {;} temporarily delete last ^E[HPg] so~
{Enter}{Right} {;} a search for ^E will result in~
{Search Left} {;} *Not Found* after the last record~

{Merge Codes}E{HPg} {;} is processed~
{Search Left}
{Backspace}{Backspace}
{Tab} {;} {Tab} is a placeholder for the deleted codes~
{CALL}action~ {;} This is where it all happens~
{Home}{Home}{Down} {;} Restore final ^E[HPg]~
{Search Left}{Tab}{Search Left}

{Backspace}
{Merge Codes}E {;} [HPg] inserted automatically~

{RETURN}

{LABEL}action~
{Home}{Home}{Up}
{DISPLAY ON}
{TEXT}0~Field number (in dBase III Date format) to be converted  (0 to quit): ~
{IF}{VAR 0}>0~

{DISPLAY OFF}
{CALL}onefield~

{ELSE}
{RETURN}

{END IF}
{GO}action~

{LABEL}delimiter~
{CHAR}1~Select delimiter character ({^V}-{^Q} {^V}/{^Q}): ~
{IF}"{VAR 1}"="/"~
{ELSE}

{IF}"{VAR 1}"="-"~
{ELSE}

{GO}delimiter~
{END IF}

{END IF}
{RETURN}

{LABEL}year~
{CHAR}2~Select number of digits for year ({^V}2{^Q} {^V}4{^Q}):~
{IF}{VAR 2}=2~
{ELSE}

{IF}{VAR 2}=4~
{ELSE}

{GO}year~
{END IF}

{END IF}
{RETURN}

{LABEL}onefield~
{ON NOT FOUND}

{RETURN}~
{ASSIGN}3~1~ {;} initial value for loop test~
{LABEL}loop~

{IF}{VAR 0}>{VAR 3}~ {;} if true, haven't reached the field yet~
{Search} {;} go to beginning of next field~

{Merge R}{Enter}
{Search}
{ASSIGN}3~{VAR 3}+1~ {;} increment the field counter~
{GO}loop~

{ELSE}
{CALL}onedate~ {;} at the right field, ready to do it~

{END IF}

{;} loop return is in body before {ELSE}~
{GO}onefield~

{LABEL}onedate~
{ON NOT FOUND} {;} means we're at the end of the file~

{RETURN NOT FND}~
{IF}{VAR 2}=2~ {;} delete century indication~

{Del}{Del}
{Right}{Right} {;} move past year~

{ELSE} {;} keep century indication~
{Esc}4{Right} {;} move past year~

{END IF}
{Block}{Esc}4{Right} {;} block month and day~
{Move}bm {;} move the block~
{Home}{Home}{Left} {;} before the year~
{Enter} {;} insert blocked month and day~
{Right}{Right}{VAR 1} {;} insert delimiter btwn month & day~
{Right}{Right}{VAR 1} {;} and between day & year~
{Search} {;} go to the end of current record~

{Merge Codes}E{HPg}
{Search}

{RETURN}

{LABEL}getfile~
{ASSIGN}5~N~
{CHAR}5~Is the correct document currently on the screen? (Y/N):  {^]}{VAR 5}

{^\}{Left}~
{IF}"{VAR 5}"="Y"~

{RETURN}
{ELSE}

{IF}"{VAR 5}"="y"~
{RETURN}

{ELSE}
{PROMPT}Retrieve document and re-run macro.~
{WAIT}20~

{END IF}
{END IF}

{QUIT}

